Öljyn määrä säiliössä

Heikki Apiola

19.1.2011

Liittyy matematiikkalehti Solmun artikkeliin: Riittääkö lämmitysöljy

http://solmu.math.helsinki.fi/2011/1/apiola.pdf

Maan sisällä makaava lieriön muotoinen säiliö.
Pohjaympyrän säde: R, Lieriön pituus: L

```
> restart: with(plots):with(linalg): # Alustuksia
```

Lieriön kuva

Avaamalla kappaleen nuolta klikkaamalla näet piirtokoodin. Kuvaa voit pyöritellä hiirellä. (Edellyttää Maplea, pdf-tiedostoa voi vain katsoa.)

```
> lierionkuva;
```



Piirretään lieriön profiiliympyrän kuva

Jos avaat kappaleen nuolta klikkaamalla, näet piirtokomennot.

```
> ympkuva;
```



Olkoon siis lieriön pituus=L ja pohjaympyrän säde=R. Kuten kuvasta nähdään ja artikkelissa tarkemmin selitetään, saadaan lieriön tilavuudelle kaava:
(Huomaa, että punaiset kaavat ovat Maple-ohjelmalle annettavia komentoja, siniset ohjelman palauttamia tuloksia.)

```
> tilavuus:=((Pi-alpha)*R^2+y*sqrt(R^2 - y^2))*L;
```

$$tilavuus := \left((\pi-\alpha)\,R^2 + y\sqrt{R^2-y^2}\right)L \tag{1}$$

Muuttujalla **tilavuus** on nyt yllä näkyvä symbolinen arvo

```
> alpha:=arccos(y/R);
```

$$\alpha := \arccos\left(\frac{y}{R}\right) \tag{2}$$

Mittatikku antaa öljyn pinnan korkeuden h pohjalta lukien, ts.

```
> y:=h-R;
```

$$y := h - R \tag{3}$$

Yllä tilavuuden lausekkeessa esiintyvät muttujat ovat "vapaita muuttujia", ts. niille ei ole asetettu arvoa. Nyt, kun annoimme siinä esiintyville muuttujille α ja y arvot,
ne sijoittuvat automaattisesti tilavuuden lausekkeeseen, no katsotaan:

```
> tilavuus;
```

$$\left(\left(\pi - \arccos\left(\frac{h-R}{R}\right)\right)R^2 + (h-R)\sqrt{R^2-(h-R)^2}\right)L \tag{4}$$

Helpottaaksemme jatkokäsittelyä, määrittelemme tilavuuden **lausekkeen** sijasta **funktioksi:**

Maple:ssa on kaksi tapaa, emme puutu tekniikoihin tarkemmin, tässä yhteydessä sopii tämä:

```
> V:=unapply(tilavuus,h);
```

$$V := h \to \left(\left(\pi - \arccos\left(\frac{h-R}{R}\right)\right) R^2 + (h-R)\sqrt{R^2-(h-R)^2}\right) L \quad \textbf{(5)}$$

Koska lieriön tilavuus = 4 m^3, *niin* $\pi R^2 L = 4$, *joten*

```
> L:=4/(Pi*R^2);
```

$$L := \frac{4}{\pi R^2} \quad \textbf{(6)}$$

```
> V(h);
```

$$\frac{4\left(\left(\pi - \arccos\left(\frac{h-R}{R}\right)\right) R^2 + (h-R)\sqrt{R^2-(h-R)^2}\right)}{\pi R^2} \quad \textbf{(7)}$$

Nyt siis myös L on sijoitettu R:n avulla lausuttuna V-funktion lausekkeeseen..
Meillä on siis tuntematon R ja sen määräämiseksi yhtälö V(0.95) = V(0.1) + 3

```
> yhtalo:=V(0.95)=3+V(0.1);
```

$$yhtalo := \frac{4\left(\left(\pi - \arccos\left(\frac{0.95-R}{R}\right)\right) R^2 + (0.95-R)\sqrt{R^2-(0.95-R)^2}\right)}{\pi R^2} = 3 + \frac{4\left(\left(\pi - \arccos\left(\frac{0.1-R}{R}\right)\right) R^2 + (0.1-R)\sqrt{R^2-(0.1-R)^2}\right)}{\pi R^2} \quad \textbf{(8)}$$

Nähdään, että vähentämällä 4 kummaltakin puolelta, voitaisiin yhtälö sieventää artikkelissa esiintyvään muotoon. Maplen komentaminen tuohon ei näytä olevan ihan helppoa, eikä sillä ole merkitystä jatkokösittelyssä, kun jatketaan Maplella.

Yhtälön ratkaiseminen

```
> with(plots):
> vasen:=lhs(yhtalo);oikea:=rhs(yhtalo);
```

$$vasen := \frac{4\left(\left(\pi - \arccos\left(\frac{0.95-R}{R}\right)\right) R^2 + (0.95-R)\sqrt{R^2-(0.95-R)^2}\right)}{\pi R^2}$$

$$oikea := 3 + \frac{4\left(\left(\pi - \arccos\left(\frac{0.1-R}{R}\right)\right) R^2 + (0.1-R)\sqrt{R^2-(0.1-R)^2}\right)}{\pi R^2} \quad \textbf{(3.1)}$$

Tutkitaan graafisesti:

```
> kappyra:=plot(vasen-oikea,R=0.5..1,thickness=2,axes=box,
  gridlines=true): rinkula:=plot([0.65],[0],style=point,symbol=
  circle,symbolsize=20,color=black):
> display([kappyra,rinkula]);
```



Zoomataan lähemmäs:

```
> kappyra:=plot(vasen-oikea,R=0.6..0.7,thickness=2,axes=box,
  gridlines=true): rinkula:=plot([0.648],[0],style=point,
  symbol=circle,symbolsize=20,color=black):
> display([kappyra,rinkula]);
```



Kursorin kohdistuksella luetaan: **R=0.648.**
Ratkaistaan vielä numeerisella yhtälön ratkaisijalla, alkupisteenä R=0.65.
(Silmämääräisesti käyrä ei juuri eroa tangentistaan tällä zoomauksella**.** Niinpä tällä alueella Newtonin menetelmän pitäisi toimia erittäin hyvin**.)**

```
> fsolve(vasen-oikea,R=0.65);
```

$$0.6480360777 \qquad \textbf{(3.2)}$$

Mittatikun asteikon laskeminen

```
> R:='R':V(h);
```

$$\frac{4\left(\left(\pi-\arccos\left(\frac{h-R}{R}\right)\right)R^2+(h-R)\sqrt{R^2-(h-R)^2}\right)}{\pi R^2} \qquad \textbf{(4.1)}$$

Annetaan R:lle yllä laskettu arvo.

```
> R:=0.648036077;
```

$$R := 0.648036077 \qquad \textbf{(4.2)}$$

```
> V(h);
```

$$\frac{1}{\pi}\Big(9.524926272\,\big(0.4199507571\,\pi - 0.4199507571\arccos(1.543123964\,h - 1.000000000) + (h - 0.648036077)\sqrt{0.4199507571 - (h - 0.648036077)^2}\big)\Big) \quad \textbf{(4.3)}$$

Nyt meillä on funktio V, joka laskee numeerisen tilavuusarvon V(h) syötteenä annetulle korkeusarvolle h. Maple täytyy komentaa laskemaan numeerinen likiarvo yhdistämällä laskentaan funktio **evalf.** Funktioiden yhdistämiseen on symboli **@.**

```
> Vf:=evalf@V;
```

$$Vf := evalf@V \quad \textbf{(4.4)}$$

Tarkistetaan, että yhtälömme toteutuu.

```
> Vf(0.95)=Vf(0.1)+3;
```

$$3.142117686 = 3.142117684 \quad \textbf{(4.5)}$$

```
> Vf(2*R);
```

$$4.00000 \quad \textbf{(4.6)}$$

Kun otettiin R:lle näin tarkkaan, yhtälön puolet eroavat vasta 10:nnessä numerossa. Toki mittauksissa on senverran epätarkkuutta, että pienempikin numeroiden määrä riittäisi hyvin. Mutta antaa nyt Maplen laskea tällä arvolla.

(Kiintoisa havainto: **Jokisen öljysäiliön** täyttöfirma oli nähtävästi tähdännyt siihen, että säiliöön laitetaan öljyä **1000π litraa,** kolmen numeron tarkkuudella onnistuivat)

Taulukon laskenta

```
> askel:=0.1; korkeudet:=[seq(k*askel,k=0..13)];  # Taulukoidaan 10 cm askelin.
```

$$askel := 0.1$$

$$korkeudet := [0., 0.1, 0.2, 0.3, 0.4, 0.5, 0.6, 0.7, 0.8, 0.9, 1.0, 1.1, 1.2, 1.3] \quad \textbf{(4.1.1)}$$

```
> tilavuudet:=map(Vf,korkeudet);
```

$$tilavuudet := [0., 0.1421176836, 0.3920175214, 0.7013129324, 1.049684262, 1.423387067, 1.811413658, 2.203974910, 2.591628343, 2.964555524, 3.311676868, 3.619088904, 3.866044676, 4.000000000 + 0.001133946276\,I] \quad \textbf{(4.1.2)}$$

```
> interface(displayprecision=5):                    # Näyttötarkkuus 5 numeroa
```

```
> tilavuudet;
```

$$[0.00000, 0.14212, 0.39202, 0.70131, 1.04968, 1.42339, 1.81141, 2.20397, 2.59163, 2.96456, 3.31168, 3.61909, 3.86604, 4.00000 + 0.00113\,I] \quad \textbf{(4.1.3)}$$

```
> with(linalg):with(LinearAlgebra):
```

```
> matrix([korkeudet,tilavuudet]);
```

$$[[0.00000, 0.10000, 0.20000, 0.30000, 0.40000, 0.50000, 0.60000, 0.70000, 0.80000, 0.90000, 1.00000, 1.10000, 1.20000, 1.30000], [0.00000, 0.14212, 0.39202, 0.70131, 1.04968, 1.42339, 1.81141, 2.20397, 2.59163, 2.96456, 3.31168, 3.61909, 3.86604, 4.00000 + 0.00113\,I]] \quad \textbf{(4.1.4)}$$

Kuriositeetti: Kun pyydetään laskemaan tilavuutta korkeusarvolla, joka menisi säiliön ulkopuolelle (1.3 > 2R), saadaan tilavuudelle imaginaariosa.

Lasketaan 1 cm:n askeleella, kuten alkuperäisessä tehtävänannossa. Toistetaan edelliset komennot muuttaen asekel ja niiden lukumäärä:

```
> askel:=0.01: korkeudet:=[seq(k*askel,k=0..130)]:
```

```
> tilavuudet:=map(Vf,korkeudet):
```

Taulukko ei mahdu vaakasuunnassa, joten käännetään se pysyyn ja jaetaan osiin. Näytetään muodossa:

[hsarake,Vsarake,hsarake,Vsarake,...,hsarake,Vsarake]

```
> Taulukko:=Transpose(Matrix([korkeudet,tilavuudet])):
```

```
> osa1:=<Taulukko(1..10,1..2)|Taulukko(11..20,1..2)|Taulukko
(21..30,1..2)|Taulukko(31..40,1..2)>;
```

$$osa1 := \begin{bmatrix} 0.00000 & 0.00000 & 0.10000 & 0.14212 & 0.20000 & 0.39202 & 0.30000 & 0.70131 \\ 0.01000 & 0.00459 & 0.11000 & 0.16356 & 0.21000 & 0.42069 & 0.31000 & 0.73465 \\ 0.02000 & 0.01296 & 0.12000 & 0.18590 & 0.22000 & 0.44993 & 0.32000 & 0.76836 \\ 0.03000 & 0.02375 & 0.13000 & 0.20910 & 0.23000 & 0.47969 & 0.33000 & 0.80242 \\ 0.04000 & 0.03647 & 0.14000 & 0.23311 & 0.24000 & 0.50997 & 0.34000 & 0.83683 \\ 0.05000 & 0.05085 & 0.15000 & 0.25788 & 0.25000 & 0.54074 & 0.35000 & 0.87156 \\ 0.06000 & 0.06669 & 0.16000 & 0.28338 & 0.26000 & 0.57198 & 0.36000 & 0.90661 \\ 0.07000 & 0.08384 & 0.17000 & 0.30957 & 0.27000 & 0.60368 & 0.37000 & 0.94196 \\ 0.08000 & 0.10219 & 0.18000 & 0.33643 & 0.28000 & 0.63581 & 0.38000 & 0.97760 \\ 0.09000 & 0.12164 & 0.19000 & 0.36392 & 0.29000 & 0.66836 & 0.39000 & 1.01351 \end{bmatrix} \quad \textbf{(4.1.5)}$$

```
> osa2:=<Taulukko(41..50,1..2)|Taulukko(51..60,1..2)|Taulukko
(61..70,1..2)|Taulukko(71..80,1..2)>;
```

$$osa2 := \begin{bmatrix} 0.40000 & 1.04968 & 0.50000 & 1.42339 & 0.60000 & 1.81141 & 0.70000 & 2.20397 \\ 0.41000 & 1.08611 & 0.51000 & 1.46171 & 0.61000 & 1.85062 & 0.71000 & 2.24312 \\ 0.42000 & 1.12278 & 0.52000 & 1.50017 & 0.62000 & 1.88987 & 0.72000 & 2.28220 \\ 0.43000 & 1.15967 & 0.53000 & 1.53875 & 0.63000 & 1.92914 & 0.73000 & 2.32122 \\ 0.44000 & 1.19678 & 0.54000 & 1.57744 & 0.64000 & 1.96842 & 0.74000 & 2.36016 \\ 0.45000 & 1.23410 & 0.55000 & 1.61624 & 0.65000 & 2.00772 & 0.75000 & 2.39901 \\ 0.46000 & 1.27161 & 0.56000 & 1.65513 & 0.66000 & 2.04701 & 0.76000 & 2.43777 \\ 0.47000 & 1.30930 & 0.57000 & 1.69410 & 0.67000 & 2.08629 & 0.77000 & 2.47642 \\ 0.48000 & 1.34717 & 0.58000 & 1.73314 & 0.68000 & 2.12555 & 0.78000 & 2.51495 \\ 0.49000 & 1.38520 & 0.59000 & 1.77225 & 0.69000 & 2.16478 & 0.79000 & 2.55336 \end{bmatrix} \quad \textbf{(4.1.6)}$$

```
> osa3:=<Taulukko(81..90,1..2)|Taulukko(91..100,1..2)
|Taulukko(101..110,1..2)|Taulukko(111..120,1..2)>;
```

$$osa3 := \begin{bmatrix} 0.80000 & 2.59163 & 0.90000 & 2.96456 & 1.00000 & 3.31168 & 1.10000 & 3.61909 \\ 0.81000 & 2.62975 & 0.91000 & 3.00063 & 1.01000 & 3.34447 & 1.11000 & 3.64695 \\ 0.82000 & 2.66772 & 0.92000 & 3.03643 & 1.02000 & 3.37686 & 1.12000 & 3.67420 \\ 0.83000 & 2.70552 & 0.93000 & 3.07196 & 1.03000 & 3.40882 & 1.13000 & 3.70080 \\ 0.84000 & 2.74315 & 0.94000 & 3.10719 & 1.04000 & 3.44034 & 1.14000 & 3.72672 \\ 0.85000 & 2.78058 & 0.95000 & 3.14212 & 1.05000 & 3.47140 & 1.15000 & 3.75194 \\ 0.86000 & 2.81782 & 0.96000 & 3.17672 & 1.06000 & 3.50198 & 1.16000 & 3.77641 \\ 0.87000 & 2.85485 & 0.97000 & 3.21100 & 1.07000 & 3.53206 & 1.17000 & 3.80011 \\ 0.88000 & 2.89165 & 0.98000 & 3.24492 & 1.08000 & 3.56162 & 1.18000 & 3.82298 \\ 0.89000 & 2.92823 & 0.99000 & 3.27849 & 1.09000 & 3.59064 & 1.19000 & 3.84497 \end{bmatrix} \quad \textbf{(4.1.7)}$$

```
> Transpose(Taulukko(121..130,1..2)); # Viimeinen osa
  vaakasuoraan.
```

[[1.20000, 1.21000, 1.22000, 1.23000, 1.24000, 1.25000, 1.26000, 1.27000, 1.28000, 1.29000], [3.86604, 3.88613, 3.90516, 3.92304, 3.93969, 3.95498, 3.96873, 3.98074, 3.99066, 3.99783]] **(4.1.8)**

```
> 2*R;
```

1.29607 **(4.7)**

Eiköhän se ole siinä!